
Program A.II.1.  A FORTRAN Program to Perform Saturation Pressure 
Search 

 
c  Output of this program: P, T, and x of a control volume       
c  Inputs to this program: v, u 
c 
      print *,’Enter > specific volume (v) in ft3/lbm and specific’, 
     1 ‘internal energy (u) in Btu/lbm’ 
      read(*,*) v,u 
      T = 344.00 
1     continue 
      n=n+1 
      call watsat(T,P,vf,vfg,vg,hf,hfg,hg,uf,ufg,ug,dpdt,dvfdt, 
     1        dvfgdt,dvgdt,dhfdt,dhfgdt,dhgdt,dufdt,dufgdt,dugdt,n) 
      Foft=(v-vf)*ufg-(u-uf)*vfg 
      Fpft=-ufg*dvfdt+(v-vf)*dufgdt+vfg*dufdt-(u-uf)*dvfgdt 
      eps=Foft/Fpft 
      T=T-eps 
      if(abs(eps).le.1E-5.or.n.gt.250) go to 2 
 
      go to 1  
2     continue       
      x=(u-uf)/ufg 
      write(*,4) 
      write(*,3) n,eps,v,u,t,p,x 
3     format(   
     1       ' Number of steps:....................',i14,/, 
     2       ' Convergence criterion:..............',E14.4,/, 
     3       ' Specific volume (ft3/lbm):..........',f14.4,/, 
     4       ' Specific int. energy (Btu/lbm):.....',f14.4,/, 
     5       ' Temperature (F):....................',f14.2,/, 
     6       ' Pressure (psia):....................',f14.2,/, 
     7       ' Quality (-):........................',f14.4,//) 
4     format(//////////////////////)       
      stop 
      end 
c---------------------------------------------------------------------- 
      subroutine watsat(t,p,vf,vfg,vg,hf,hfg,hg,uf,ufg,ug,dpdt,dvfdt, 
     1            dvfgdt,dvgdt,dhfdt,dhfgdt,dhgdt,dufdt,dufgdt,dugdt,n) 
c 
c  Water properties p, vf, vfg, uf, and ufg are correlated to saturated 
c  tempearture between 32 and 656 Deg F (0.0886 < P < 2300 psia).  
c  Data in British units.   
c  
      data c/.1850899/ 
      if(t.lt.32.0.or.t.gt.656.0) go to 1 
         call psatot(t,p,dpdt) 
         call vfsatt(t,vf,dvfdt) 
         call vfgsat(t,vfg,dvfgdt) 
         call ufsatt(t,uf,dufdt) 
         call ugsatt(t,ug,dugdt) 
         vg=vf+vfg 
         dvgdt=dvfdt+dvfgdt 
         ufg=ug-uf 
         dufgdt=dugdt-dufdt 
         hf=uf+c*p*vf 



         hg=ug+c*p*vg 
         hfg=hg-hf 
         dhfdt=dufdt+c*(dpdt*vf+p*dvfdt) 
         dhgdt=dugdt+c*(dpdt*vg+p*dvgdt) 
         dhfgdt=dhfdt+dhgdt 
      return 
1     continue 
      write(*,*) n 
      write(*,2) t 
2     format(' Sat. Temp. Outside of the Range of Properties.',f10.3) 
      end 
c 
      subroutine psatot(tsat,psat,dpdt) 
      dimension b(8) 
 
      data a,b(1),b(2),b(3),b(4),b(5),b(6),b(7),b(8)/.6962719, 
     1     -.3579628e-1,.7568188e-3,-.6924369e-5,.4086826e-7, 
     2     -.1124099e-9,.2708745e-12,-.3145232e-15,.1427914e-18/ 
c 
      arg=0.0 
      argp=0.0 
      do 1 i=1,8 
        arg=arg+b(i)*(tsat**i) 
        argp=argp+i*b(i)*(tsat**(i-1)) 
1     continue 
      psat=a+arg 
      dpdt=argp 
      return 
      end 
c 
      subroutine vfsatt(tsat,vf,dvfdt) 
      dimension b(8) 
      data a,b(1),b(2),b(3),b(4),b(5),b(6),b(7),b(8)/62.11036, 
     1       .1722592e-1,-.2852652e-3,.1485193e-5,-.6087479e-8, 
     2       .1660097e-10,-.2821841e-13,.2669788e-16,-.1077666e-19/ 
c 
      arg=0.0 
      argp=0.0 
      do 1 i=1,8 
        arg=arg+b(i)*(tsat**i) 
        argp=argp+i*b(i)*(tsat**(i-1)) 
1     continue 
      vf=1./(a+arg) 
      dvfdt=-vf*vf*argp 
      return 
      end 
c 
      subroutine vfgsat(tsat,vfg,dvfgdt) 
      dimension b(10) 
      data a,b(1),b(2),b(3),b(4),b(5),b(6),b(7),b(8),b(9),b(10)/ 
     1     9.429309,-.4471527e-1,.1089104e-3,-.1994522e-6, 
     2     -.4738610e-10,.2567196e-11,-.1197328e-13,.3006035e-16, 
     3     -.4408149e-19,.3543467e-22,-.1208486e-25/ 
c 
      arg=0.0 
      argp=0.0 
      do 1 i=1,10 



        arg=arg+b(i)*(tsat**i) 
        argp=argp+i*b(i)*(tsat**(i-1)) 
1     continue 
      vfg=exp(a+arg) 
      dvfgdt=vfg*argp 
      return 
      end 
c 
      subroutine ufsatt(tsat,uf,dufdt) 
      dimension b(8) 
      data a,b(1),b(2),b(3),b(4),b(5),b(6),b(7),b(8)/-31.33822, 
     1    .9664909,.6886926e-3,-.7099748e-5,.3951334e-7,-.1246218e-9, 
     2    .2283063e-12,-.2246668e-15,.9254311e-19/ 
c 
      arg=0.0 
      argp=0.0 
      do 1 i=1,8 
        arg=arg+b(i)*(tsat**i) 
        argp=argp+i*b(i)*(tsat**(i-1)) 
1     continue 
      uf=a+arg 
      dufdt=argp 
      return 
      end 
c 
      subroutine ugsatt(tsat,ug,dugdt) 
      dimension b(8) 
      data a,b(1),b(2),b(3),b(4),b(5),b(6),b(7),b(8)/1016841., 
     1     920.3534,-4.762259,.4607905e-1,-.2479441e-3,.7584943e-6, 
     2     -.1346369e-8,.1283154e-11,-.5100722e-15/ 
c 
      arg=0.0 
      argp=0.0 
      do 1 i=1,8 
        arg=arg+b(i)*(tsat**i) 
        argp=argp+i*b(i)*(tsat**(i-1)) 
1     continue 
      ug=sqrt(arg+a) 
      dugdt=0.5*argp/ug 
      return 
      end 
c........................................................ 
 
 


